

Tin: A Tcl Package Manager

Version 2.1

Alex Baker

<https://github.com/ambaker1/Tin>

March 2, 2025

Abstract

Tin is a Tcl package that installs Tcl packages and their dependencies directly from Git repositories. Additionally, Tin provides various utilities for Tcl package ecosystem development.

The Tin List

The “Tin List” is a database of package version installation instructions, consisting of the package name, version, Git repository, release tag, and installation file path within the repository. It tells the Tcl interpreter which release of a repository to clone, and which file within that release to run to install the specified version of the package.

Modifying the Tin List

The command *tin add* adds package installation entries to the Tin List.

```
tin add $name $version $repo $tag $file
```

| | |
|------------------|------------------------------------|
| \$name | Package name. |
| \$version | Package version. |
| \$repo | Git repository URL. |
| \$tag | Git release tag for version. |
| \$file | Installer file path in repository. |

Package installation entries can be removed from the Tin List in a session with the command *tin remove*, and the command *tin clear* removes all entries.

```
tin remove $name <$version> <$repo>
```

```
tin clear
```

| | |
|------------------|---|
| \$name | Package name. |
| \$version | Package version to remove (optional, default all versions). |
| \$repo | Repository to remove (optional, default all repositories). |

Example 1: Adding entries to the Tin List

Code:

```
tin add foo 1.0 https://github.com/username/foo v1.0 install_foo.tcl
tin add foo 1.1 https://github.com/username/foo v1.1 install_foo.tcl
tin add foo 2a0 https://github.com/username/foo v2a0 install_foo.tcl
tin add foo 2.0 https://github.com/username/foo v2.0 install_foo.tcl
tin add foo 2.0.1 https://github.com/username/foo v2.0.1 install_foo.tcl
```

Auto-Populating the Tin List

The command `tin autoadd` uses specifically formatted Git tags corresponding to package versions to automatically add entries to the Tin List. The optional version requirements arguments can be used to import a range of versions from a repository. Returns unsorted list of versions added.

```
tin autoadd $name $repo $file <<-exact> $version> <$reqs ...>
```

| | |
|-------------------|--|
| \$name | Package name. |
| \$repo | Remote Git repository URL. |
| \$file | Installer file path in repo. |
| \$version | Package version (-exact specifies exact version). |
| \$reqs ... | Package version requirements, mutually exclusive with -exact option. |

Package version requirements can take the form “min”, “min-”, and “min-max”.

- “min”: Min-bounded. Must be greater than or equal to “min”, but the same major version as “min”.
- “min-”: Min-unbounded. Must be greater than or equal to “min”.
- “min-max”: Bounded. Must be greater than or equal to “min”, and less than “max”.

Example 2: Auto-adding entries to the Tin List

Code:

```
tin autoadd foo https://github.com/username/foo install_foo.tcl 1.5-
```

Auto-Add Tag Pattern

In order for package installation entries to be added with auto-add, the Git repository must have tags corresponding directly with the package versions, such as “v1.2.3”.

To be specific, tags must match the following regex pattern:

```
^v(0|[1-9]\d*)(\.(0|[1-9]\d*))*([ab](0|[1-9]\d*)(\.(0|[1-9]\d*))*)?>
```

Note that this is not the same as “SemVer” (<https://semver.org/>), which is the standard version number format. This is because the auto-add tag pattern specifically matches all version numbers compatible with Tcl, with a prefix of “v”. Most notably, the format for alpha and beta versions is different in Tcl, where “a” and “b” replace one of the periods (e.g. “v2a0”), and effectively represent “.-2.” and “.-1.”, respectively.

Accessing the Tin List

The command *tin get* queries basic information about the Tin List, and returns blank if the requested entry does exist. Returns a dictionary associated with the supplied arguments.

```
tin get $name <$version> <$repo>
```

| | |
|------------------|---------------------|
| \$name | Package name. |
| \$version | Package version. |
| \$repo | Git repository URL. |

The available packages in the Tin List can be queried with the command *tin packages*.

```
tin packages <$pattern>
```

| | |
|------------------|--|
| \$pattern | Optional “glob” pattern, default “*”, or all packages. |
|------------------|--|

An unsorted list of available versions for a Tin package that satisfy version requirements can be queried with the command *tin versions*. Similarly, the command *tin available* returns the version that would be installed with *tin install*, or blank if none can be found.

```
tin versions $name <<-exact> $version> <$reqs ...>
```

```
tin available $name <<-exact> $version> <$reqs ...>
```

| | |
|-------------------|--|
| \$name | Package name. |
| \$version | Package version (-exact specifies exact version). |
| \$reqs ... | Package version requirements, mutually exclusive with -exact option. |

Example 3: Getting info from the Tin List

Code:

```
package require tin
tin add foo 1.0 https://github.com/username/foo v1.0 install_foo.tcl
puts [tin get foo]
```

Output:

```
1.0 {https://github.com/username/foo {v1.0 install_foo.tcl}}
```

Installing and Uninstalling Packages

The command *tin install* installs packages directly from a Git repository, and returns the version installed.

The command *tin depend* installs packages only if they are not installed, and returns the version number installed (useful for dependencies in installation scripts).

The command *tin uninstall* uninstalls packages (as long as they are in the Tin List), and returns blank if successful, or error if it was unsuccessful in uninstalling the package.

```
tin install $name <<-exact> $version> <$reqs ...>
```

```
tin depend $name <<-exact> $version> <$reqs ...>
```

```
tin uninstall $name <<-exact> $version> <$reqs ...>
```

| | |
|-------------------|--|
| \$name | Package name. |
| \$version | Package version (-exact specifies exact version). |
| \$reqs ... | Package version requirements, mutually exclusive with -exact option. |

By default, uninstalling a package simply deletes the library folder associated with the package. However, if a “pkgUninstall.tcl” file is located within the package folder, it will run that file instead, with the variable `$dir` set to the package library folder path, similar to how “pkgIndex.tcl” files work.

Example 4: Complex uninstall file “pkgUninstall.tcl”

Code:

```
set bindir [file dirname [info nameofexecutable]]
file delete [file join $bindir foo.bar]; # delete associated file in the bin directory
file delete -force $dir; # Clean up package
```

Upgrading Packages

Upgrading packages involves first installing the new version with *tin install*, then uninstalling the old version with *tin uninstall*. The command *tin check* returns an upgrade list of available minor and patch upgrades for packages. The command *tin upgrade* checks for available upgrades with *tin check*, and then upgrades the packages, returning the upgrade list. If there is no upgrade available, the upgrade list will be empty. If the *-all* option is specified, the format of the upgrade list is “name {old new ...} ...”. Otherwise, the format of the upgrade list is “old new”.

```
tin check $name <<-exact> $version> <$reqs ...>
tin check -all <$names>
```

```
tin upgrade $name <<-exact> $version> <$reqs ...>
tin upgrade -all <$names>
```

| | |
|-------------------|--|
| -all | Option to look for upgrades in all installed major versions of the packages. |
| \$names | Package names. Default searches all packages in the Tin List. |
| \$name | Package name. |
| \$version | Package version (-exact specifies exact version). |
| \$reqs ... | Package version requirements, mutually exclusive with -exact option. |

Example 5: Upgrading Tin

Code:

```
# Upgrade Tin
package require tin
tin autoadd tin https://github.com/ambaker1/Tin install.tcl
tin upgrade tin
```

Loading and Importing Packages

The command *tin require* is similar to the Tcl command *package require*, but with the added feature that if the package is missing, it will try to install it with *tin install*.

The command *tin import* additionally handles most use-cases of *namespace import*. Both *tin require* and *tin import* return the version number of the package imported.

```
tin require $name <<-exact> $version> <$reqs ...>
```

| | |
|-------------------|--|
| \$name | Package name. |
| \$version | Package version (-exact specifies exact version). |
| \$reqs ... | Package version requirements, mutually exclusive with -exact option. |

```
tin import <-force> <$patterns from> $name <<-exact> $version> <$reqs ...> <as $ns>
```

| | |
|-------------------|---|
| -force | Option to overwrite existing commands. |
| \$patterns | Commands to import, or “glob” patterns, default “*”, or all commands. |
| \$name | Package name. |
| \$version | Package version (-exact specifies exact version). |
| \$reqs ... | Package version requirements, mutually exclusive with -exact option. |
| \$ns | Namespace to import into. Default global namespace, or “::”. |

Example 6: Importing all commands from package “foo”, version 1.0

Code:

```
package require tin
tin import foo 1.0
```

Generic Package Utilities

Tin works alongside the Tcl *package* commands, and provides a few package utility commands that do not interface with the Tin List.

Check if a Package is Installed

The command *tin installed* returns the package version number that would be loaded with *package require*, or blank if the package is not installed. If there is no package version in the Tcl package database satisfying the requirements, it will call the *package unknown* script to load *package ifneeded* statements from “pkgIndex.tcl” files, just like what *package require* does, but without loading the package.

```
tin installed $name <<-exact> $version> <$reqs ...>
```

| | |
|-------------------|--|
| \$name | Package name. |
| \$version | Package version (-exact specifies exact version). |
| \$reqs ... | Package version requirements, mutually exclusive with -exact option. |

Unload a Package/Namespace

The command *tin forget* is short-hand for both *package forget* and *namespace delete*, but it will not throw an error if there is no namespace corresponding with the package name. It is especially useful for reloading packages within an instance of Tcl.

```
tin forget $name ...
```

| | |
|---------------|---------------|
| \$name | Package name. |
|---------------|---------------|

Example 7: Loading and reloading a package

Code:

```
package require foo
tin forget foo
package require foo
```

Utilities for Package Development

In addition to commands for installing and loading packages, Tin provides a few commands intended to help in writing installation and build files for your packages.

Creating Package Directories

The command *tin mkdir* creates a library directory to install a package in, with a normalized naming convention that allows it to be uninstalled easily with *tin uninstall*.

```
tin mkdir <-force> <$basedir> $name $version
```

| | |
|------------------|--|
| -force | Option to create fresh library directory (deletes existing folder). |
| \$basedir | Base directory, default searches Tcl <code>auto_path</code> to find a writable path. |
| \$name | Package name. |
| \$version | Package version. |

See the example installation file for a package “foo” that requires the package “bar 1.2”, and installs in library folder “foo-1.0”.

Example 8: Example file “install_foo.tcl”

Code:

```
package require tin
tin add bar 1.2 https://github.com/username/bar v1.2 install_bar.tcl
tin depend bar 1.2
set dir [tin mkdir -force foo 1.0]
file copy README.md LICENSE foo.tcl pkgIndex.tcl $dir
```

Building Library Files from Source with Configuration Variable Substitution

The command *tin bake* takes an input text file, and writes an output text file after substitution of configuration variables such as @VERSION@. This is helpful for ensuring that the package version is consistent across the entire project. If a source directory is used as input, it will “batch bake” all “.tin” files.

```
tin bake $src $target $config
tin bake $src $target $varName $value ...
```

| | |
|------------------------------|--|
| \$src | Source file, or directory with “.tin” files. |
| \$target | Target file, or directory to write “.tcl” files to. |
| \$config | Dictionary of config variable names and values. Config variables must be uppercase and alphanumeric. |
| \$varName \$value ... | Config variable names and values. Mutually exclusive with \$config. |

See below for an example of how *tin bake* can be used to automatically update a “pkgIndex.tcl” file:

Example 9: Building a “pkgIndex.tcl” file

Code:

```
package require tin
tin bake pkgIndex.tin pkgIndex.tcl {VERSION 1.0}
```

Output:

"pkgIndex.tin" :

```
package ifneeded foo @VERSION@ [list source [file join $dir foo.tcl]]
```

"pkgIndex.tcl" :

```
package ifneeded foo 1.0 [list source [file join $dir foo.tcl]]
```

Basic Unit Testing

The command *assert* can be used for basic unit testing of Tcl scripts. It throws an error if the statement is false. Otherwise, it simply returns nothing and the script continues.

There are two forms of this command, one which passes input through the Tcl *expr* command, and the other which does value comparison.

```
tin assert $expr <$message>
tin assert $value $op $expected <$message>
```

| | |
|-------------------|--|
| \$expr | Tcl math expression to evaluate as boolean. |
| \$value | Value to compare. |
| \$op | Comparison operator, or “is” to assert type. |
| \$expected | Comparison value. |
| \$message | Optional error message to add context to assertion error. Default blank. |

Example 10: Asserting values and types

Code:

```
tin assert {2 + 2 == 4}; # Asserts that math works
tin assert 5.0 is double; # Asserts that 5.0 is indeed a number
tin assert {hello world} is integer; # This is false
```

Output:

```
expected integer value but got "hello world"
```

Example 11: Provide context to invalid procedure input

Code:

```
proc subtract {x y} {
    tin assert $x > $y {x must be greater than y}
    expr {$x - $y}
}
subtract 2.0 3.0
```

Output:

```
x must be greater than y
assert 2.0 > 3.0 failed
while executing
"subtract 2.0 3.0"
```

For more advanced unit testing, the built-in *tcltest* package is recommended.

Note about installing GitHub release assets

The architecture of Tin is rather simple: it clones a repository using a specified tag, and runs a Tcl file within that cloned repository that installs the package. If you are using GitHub to release your code, you may have release assets, such as binary files or zipped folders, that you wish to install. In this case, you can use GitHub CLI (<https://cli.github.com/>) in your installation file to add these compiled assets.

Example 12: Advanced installation file

Code:

```
# Install dependency
package require tin
tin add bar 1.2 https://github.com/username/bar v1.2 install_bar.tcl
tin depend bar 1.2

# Use GitHub CLI to download release assets (InstallFoo1.0.exe and foo1.0.zip)
exec gh release download v1.0 --clobber; # --clobber overwrites

# Customize installation for operating system
switch -- $::tcl_platform(os) {
    Windows {
        # Run binary installation file
        exec InstallFoo1.0.exe
    }
    default {
        # Extract zip folder, and install contents
        package require zipfile::decode 0.7.1
        ::zipfile::decode::unzipfile foo1.0.zip [pwd]
        set dir [tin mkdir -force foo 1.0]
        set files [glob -directory foo1.0 *]
        file copy {*}$files $dir
    }
}
```

Command Index

tin add, 2
tin assert, 11
tin autoadd, 3
tin available, 4
tin bake, 10
tin check, 6
tin clear, 2
tin depend, 5
tin forget, 8
tin get, 4
tin import, 7
tin install, 5
tin installed, 8
tin mkdir, 9
tin packages, 4
tin remove, 2
tin require, 7
tin uninstall, 5
tin upgrade, 6
tin versions, 4